

# Number Play, Calculators, and Card Tricks: Mathemagical Black Holes

Michael W. Ecker

The legend of Sisyphus is a lesson in inevitability. No matter how Sisyphus tried, the small boulder he rolled up the hill would always come down at the last minute, pulled inexorably by gravity.

Like the legend, the physical universe has strange entities called black holes that pull everything toward them, never to escape. But did you know that we have comparable bodies in recreational mathematics?

At first glance, these bodies may be even more difficult to identify in the world of number play than their more famous brethren in physics. What, after all, could numbers such as 123, 153, 6174, 4, and 15 have in common with each other, as well as with various card tricks?

These are mathematical delights interesting in their own right, but much more so collectively because of the common theme linking them all. I call such individual instances *mathemagical black holes*.

## The Sisyphus String: 123

Suppose we start with any natural number, regarded as a string, such as 9,288,759. Count the number of even digits, the number of odd digits, and the total number of digits. These are 3 (three evens), 4 (four odds), and 7 (seven is the total number of digits), respectively. So, use these digits to form the next string or number, 347.

Now repeat with 347, counting evens, odds, total number, to get 1, 2, 3, so write down 123. If we repeat with 123, we get 123 again. The number 123 with respect to this process and the universe of numbers is a mathemagical black hole. All numbers in this universe are drawn to 123 by this process, never to escape.

---

Based on articles appearing in *(REC) Recreational and Educational Computing*.

But will every number really be sent to 123? Try a really big number now, say 12233344445555566666777777888888999999999 (or pick one of your own).

The numbers of evens, odds, and total are 20, 25, and 45, respectively. So, our next iterate is 202,545, the number obtained from 20, 25, 45. Iterating for 202,545 we find 4, 2, and 6 for evens, odds, total, so we have 426 now. One more iteration using 426 produces 303, and a final iteration from 303 produces 123.

At this point, any further iteration is futile in trying to get away from the black hole of 123, since 123 yields 123 again. If you wish, you can test a lot more numbers more quickly with a computer program in BASIC or other high-level programming language. Here's a fairly generic one (Microsoft BASIC):

```

1 CLS
2 PRINT "The 123 Mathematical Black Hole / (c) 1993, Dr. M. W. Ecker"
3 PRINT: PRINT "I'll ask you to input a positive whole number now."
4 PRINT "I'll count the numbers of even digits, odd digits, and total."
5 PRINT "From that I'll form the next number. Surprisingly, we always"
6 PRINT "wind up reaching the mathematical black hole of 123...": PRINT
7 FOR DL=1 TO 1000:NEXT
10 INPUT "What is your initial whole number"; N$: PRINT
20 IF VAL(N$) < 1 OR VAL(N$) < > INT(VAL(N$)) THEN 10
30 FOR DIGIT = 1 TO LEN(N$)
40 D$ = MID$(N$,DIGIT,1)
50 IF D$ = " " THEN 70
60 IF VAL(D$)/2 = INT(VAL(D$)/2) THEN EVEN = EVEN + 1 ELSE ODD = ODD + 1
70 NEXT DIGIT
80 PRINT "EVEN, ODD, TOTAL"
90 NU$ = STR$(EVEN) + STR$(ODD) + STR$(EVEN + ODD)
100 PRINT EVEN;" ";ODD;" ";EVEN + ODD;"---> New number is"; VAL(NU$)
110 PRINT:IF VAL(NU$) = VAL(N$) THEN PRINT "Done.": END
120 N$ = NU$: EVEN = 0: ODD = 0: GOTO 30

```

If you wish, modify line 110 to allow the program to start again. Or revise the program to automate the testing for all natural numbers in some interval.

### What Is a Mathmagical Black Hole?

There are two key features that make our example interesting:

1. Once you hit 123, you never get out, just as reaching a black hole of physics implies no escape.

2. Every element subject to the force of the black hole (the process applied to the chosen universe) is eventually pulled into it. In this case, sufficient iteration of the process applied to any starting number must eventually result in reaching 123.

Of course, once drawn in per point 2, an element never escapes, as point 1 ensures.

A *mathemagical black hole* is, loosely, any number to which other elements (usually numbers) are drawn by some stated process. Though the number itself is the star of the show, the real trick is in finding interesting processes.

**Formalized Definition.** In mathematical terms, a black hole is a triple  $(b, U, f)$ , where  $b$  is an element of a set  $U$  and  $f: U \rightarrow U$  is a function, all satisfying:

1.  $f(b) = b$ .
2. For each  $x$  in  $U$ , there exists a natural number  $k$  satisfying  $f^k(x) = b$ .

Here,  $b$  plays the role of the black-hole element, and the superscript indicates  $k$ -fold (repeated) composition of functions.

For the Sisyphus String<sup>1</sup>,  $b = 123$ ,  $U = \{\text{natural numbers}\}$ , and  $f(\text{number}) =$  the number obtained by writing down the string counting # even digits of number, # odd digits, total # digits.

Why does this example work, and why do most mathemagical black holes occur? My argument is to show that large inputs have smaller outputs, thus reducing an infinite universe to a manageable finite one. At that point, an argument by cases, or a computer check of the finitely many cases, suffices.

In the case of the 123 hole, we can argue as follows: If  $n > 999$ , then  $f(n) < n$ . In other words, the new number that counts the digits is smaller than the original number. (It's intuitively obvious, but try induction if you would like rigor.) Thus, starting at 1000 or above eventually pulls one down to under 1000. For  $n < 1000$ , I've personally checked the iterates of  $f(n)$  for  $n = 1$  to 999 by a computer program such as the one above. The direct proof is actually faster and easier, as a three-digit string for a number must have one of these possibilities for (# even digits, # odd digits, total # digits):

- (0, 3, 3)
- (1, 2, 3)

---

<sup>1</sup>For generalized sisyphian strings, see *REC*, No. 48, Fall 1992.

(2, 1, 3)  
(3, 0, 3)

So, if  $n < 1000$ , within one iteration you must get one of these four triples. Now apply the rule to each of these four and you'll see that you always produce (1, 2, 3) – thus resulting in the claimed number of 123.

### Words to Numbers: 4

Here is one that master recreationist Martin Gardner wrote to tell me about several years ago. Take any whole number and write out its numeral in English, such as FIVE for the usual 5. Count the number of characters in the spelling. In this case, it is 4 – or FOUR. So, work now with the 4 or FOUR. Repeat with 4 to get 4 again.

As another instance, try 163. To avoid ambiguity, I'll arbitrarily say that we will include spaces and hyphens in our count. Then, 163 appears as ONE HUNDRED SIXTY-THREE for a total count of 23. In turn, this gives 12, then 6, then 3, then 5, and finally 4.

Though this result is clearly language-dependent, other natural languages may have a comparable property, but not necessarily with 4 as the black hole.

### Narcissistic Numbers: 153

It is well known that, other than the trivial examples of 0 and 1, the only natural numbers that equal the sum of the cubes of their digits are 153, 370, 371, and 407. Of these, just one has a black-hole property.

To create a black hole, we need to define a universe (set  $U$ ) and a process (function  $f$ ). We start with any positive whole number that is a multiple of 3. Recall that there is a special shortcut to test whether you have a multiple of 3. Just add up the digits and see whether that sum is a multiple of 3. For instance, 111,111 (six ones) is a multiple of 3 because the sum of the digits, 6, is. However, 1,111,111 (seven ones) is not.

Since we are going to be doing some arithmetic, you may wish to take out a hand calculator and/or some paper. Write down your multiple of 3. One at a time, take the *cube* of each digit. Add up the cubes to form a new number. Now repeat the process. You must reach 153. And once you reach 153, one more iteration just gets you 153 again.

Let's test just one initial instance. Using the sum of the cubes of the digits, if we start with 432 — a multiple of 3 — we get 99, which leads to 1458, then 702, which yields 351, finally leading to 153, at which point future iterations keep producing 153. Note also that this operation or process preserves divisibility by 3 in the successive numbers.

```

10 CLS
20 PRINT "The mathematical black hole 153...": PRINT
30 PRINT "Simple test program for sum of cubes of digits...": PRINT
40 PRINT "Copyright 1993, Dr. M. W. Ecker"
50 PRINT: FOR DELAY = 1 TO 2000: NEXT
60 INPUT "Number to be tested";N: PRINT
70 IF N/3 > INT(N/3) OR N < 1 THEN PRINT "Positive multiples of 3 only!":
   GOTO 70
80 IF N > 10000000# THEN PRINT "Let's stick to numbers under 10000000":
   GOTO 80
90 N$ = STR$(N): L = LEN(N$): 'Convert to a string to manipulate digits
100 SUMCUBES = 0: 'Initialize sum
110 FOR DIGIT = 1 TO L
120 V(DIGIT) = VAL(MID$(N$,DIGIT,1)): 'Get value of each digit
130 CU(DIGIT) = V(DIGIT)*V(DIGIT)*V(DIGIT): 'Cube each digit
140 SUMCUBES = SUMCUBES + CU(DIGIT): 'Keep running total of sum of cubes
150 NEXT DIGIT
160 PRINT "The sum of the cubes of the digits of " N " is " SUMCUBES
170 IF SUMCUBES = N THEN PRINT "Success! Found a black hole!": PRINT:
   GOTO 70
180 N = SUMCUBES: GOTO 110

```

This program continues forever, so break out after you've grown weary. One nice thing is that it is easy to edit this program to test for black holes using larger powers. (It is well known that none exists for the sum of the *squares* of the digits, as one gets cycles.)

In more formal language, we obtain the 153 mathematical black hole by letting  $U = 3Z^+ = \{\text{all positive integral multiples of } 3\}$  and  $f(n) = \text{the sum of the cubes of the digits of } n$ . Then  $b = 153$  is the unique black-hole element. (For a given universe, if a black hole exists, it is necessarily unique.)

Not incidentally, this particular result, without the "black hole" terminology or perspective, gets discovered and re-discovered annually, with a paper or problem proposal in one of the smaller math journals every few years.

The argument for why it works is similar to the case with the 123 example. First of all,  $1^3 + 3^3 + 5^3 = 153$ , so 153 is indeed a fixed point. Second, for the black-hole attraction, note that, for large numbers  $n$ ,

$f(n) < n$ . Then, for suitably small numbers, by cases or computer check, each value eventually is “pulled” into the black hole of 153. I’ll omit the proof.

To find an analogous black hole for larger powers (yes, there are some), you will need first to discover a number that equals the sum of the fourth (or higher) power of its digits, and then test to see whether other numbers are drawn to it.

## Card Tricks, Even

Here’s an example that sounds a bit different, yet meets the two criteria for a black hole. It’s a classic card trick.

Remove 21 cards from an ordinary deck. Arrange them in seven horizontal rows and three vertical columns. Ask somebody to think of one of the cards without telling you which card he (or she) is thinking of.

Now ask him (or her) which of the three columns contains the card. Regroup the cards by picking up the cards by whole columns intact, *but be sure to sandwich the column that contains the chosen card between the other two columns*. Now re-lay out the cards by laying out by rows (i.e., laying out three across at a time). Repeat asking which column, regrouping cards with the designated column being in the middle, and re-dealing out by rows. Repeat one last time.

At the end, the card chosen must be in the center of the array, which is to say, card 11. This is the card in the fourth row and second column.

There are two ways to prove this, but the easier way is to draw a diagram that illustrates where a chosen card will end up next time. But for those who enjoy programs, try this one from one of my readers.

```

10 CLS: PRINT "The 21 Cards - Program by Sally Frazza
20 PRINT "adapted by Dr. M. W. Ecker for REC
100 DEFINT A-Z
110 DIM V(21), X(7,3)
120 ITER = 0
130 FOR N = 1 TO 21: V(N) = N: NEXT N
140 PRINT: PRINT "Pick a card, please...": PRINT
150 IF ITER = 3 THEN PRINT: PRINT "Your card is"; V(11): END
160 ITER = ITER + 1
170 N = 0
180 FOR I = 1 TO 7: FOR J = 1 TO 3
190 N = N + 1

```

```

200 X(I,J) = V(N)
210 NEXT J: NEXT I
220 PRINT: PRINT
221 FOR I = 1 TO 7
222 PRINT USING "##"; X(I,1); X(I,2); X(I,3)
223 NEXT I
230 PRINT
240 INPUT "Column of card (1, 2, or 3)"; C
250 IF C < 1 OR C > 3 THEN 240
260 FOR K = 1 TO 3: O(K) = K: NEXT K
270 IF C < > 2 THEN SWAP O(C), O(2)
290 N = 0
300 FOR K = 1 TO 3: FOR I = 1 TO 7
310 J = O(K)
320 N = N + 1
330 V(N) = X(I,J)
340 NEXT I: NEXT K
350 GOTO 150

```

Perhaps it is not surprising, but this trick, as with the sisyphian strings, generalizes somewhat.<sup>2</sup>

### Kaprekar's Constant: What a Difference 6174 Makes!

Most black holes, nonetheless, involve numbers. Take any four-digit number except an integral multiple of 1111 (i.e., don't take one of the nine numbers with four identical digits). Rearrange the digits of your number to form the largest and smallest strings possible. That is, write down the largest permutation of the number, the smallest permutation (*allowing initial zeros as digits*), and subtract. Apply this same process to the difference just obtained. Within the total of seven steps, you always reach 6174. At that point, further iteration with 6174 is pointless:  $7641 - 1467 = 6174$ .

*Example:* Start with 8028. The largest permutation is 8820, the smallest is 0288, and the difference is 8532. Repeat with 8532 to calculate  $8532 - 2358 = 6174$ . Your own example may take more steps, but you will reach 6174.

---

<sup>2</sup>REC, No. 48, Fall 1992.

### The Divisive Number 15

Take any natural number larger than 1 and write down its divisors, including 1 and the number itself. Now take the sum of the digits of these divisors. Iterate until a number repeats.

The black-hole number this time is 15. Its divisors are 1, 3, 5, and 15, and these digits sum to 15. This one is a bit more tedious, but it is also that much more strange at the same time. This one may defy not only your ability to explain it, but your very equilibrium.

### Fibonacci Numbers: Classic Results as Black Holes

Many an endearing problem has charmed mathophiles with the Fibonacci numbers: 1, 1, 2, 3, 5, 8, 13, 21, 34, 55, 89, . . . . The first two numbers are each 1 and successive terms are obtained by adding the immediately preceding two elements. More formally,  $F(1) = 1 = F(2)$ , and  $F(n) = F(n-1) + F(n-2)$  for integers  $n > 2$ .

One of the classic results is that the ratio of consecutive terms has a limiting value. That is, form the ratios  $F(n+1)/F(n)$ :  $\frac{1}{1} = 1\frac{2}{1} = 2\frac{3}{2} = 1.5$ ,  $\frac{5}{3} = 1.667$  (approx.),  $\frac{8}{5} = 1.6$ , etc. The ratios seem to be converging to a number around 1.6 or so. In fact, it is well known that the sequence converges to  $(1 + \sqrt{5})/2$  or the golden number, phi. Its value is roughly 1.618 . . . .

Had we used the ratios  $F(n)/F(n+1)$  instead, we would have obtained the reciprocal of phi as the limit, but fewer authors use that approach.

To get a quick and dirty feel for why phi arises, but without using a program, *assume* that the numbers  $F(n+1)/F(n)$  approach *some* limit – call it  $L$  – as  $n$  increases. Divide both sides of the equation  $F(n) = F(n-1) + F(n-2)$  by  $F(n-1)$ . For large  $n$  this equation is approximately the same as  $L = 1 + 1/L$ . If we multiply both sides by  $L$  we obtain a quadratic equation:

$$(*) \quad L^2 - L - 1 = 0.$$

Solving by the quadratic formula yields two solutions, one of which is phi. More about the second solution in a moment.

Notice that the above plausibility argument did not use the values  $F(1)$  and  $F(2)$ . Indeed, more generally, if you take any additive sequence (any

sequence — no matter what the first two terms — in which the third term and beyond are obtained by adding the preceding two), one gets the same result: convergence to phi. The first two numbers need not even be whole numbers or positive. This, too, is easy to test in a program that you can write yourself.

Thus, if we extend the definition of black hole to require only that the iterates get closer and closer to one number, we have a black hole once again.

But there is still another black hole one can derive from this. Consider the function  $f(x) = 1 + 1/x$  for nonzero real numbers  $x$ . I selected (or rather stumbled across) this function  $f$  because of the simple argument above that gave phi as a limit. Start with a seed number  $x$  and iterate function values  $f(x), f(f(x)),$  etc. One obtains convergence to phi — but still no appearance of the second number that is the solution to the quadratic equation above.

Is there a connection between the two solutions of the quadratic equation? First, each number is the negative reciprocal of the other. Each is also one minus the other. Second, had we formed the ratios  $F(n)/F(n + 1)$ , we would have obtained the absolute value of the second solution instead of the first solution.

Third, note that our last function  $f, f(x) = 1 + 1/x$ , could not use an input of 0 because division by zero is undefined. The solution to the equation  $f(x) = 1 + 1/x = 0$  is just  $x = -1$ . Thus,  $x$  may not equal  $-1$  either.

We have not finished. We must now avoid  $f(x) = -1$  (otherwise,  $f(f(x)) = 0$  and then  $f(f(f(x)))$  is undefined). Solving  $1 + 1/x = -1$  gives  $x = -1/2$ . If we continue now working backward with function  $f$ 's preimages we find, in succession, that we must similarly rule out  $-2/3$ , then  $-3/5$ ,  $-5/8$ ,  $-8/13$ , etc. Notice that these fractions are precisely the negatives of the ratios of consecutive Fibonacci numbers in the reverse order that we considered. All of these must be eliminated from the universe for  $f$ , along with 0. The limiting ratio, the second golden ratio, must also be eliminated from the universe.

In closing out this brief connection of Fibonacci numbers to the topic, I would be remiss if I did not follow my own advice on getting black holes by looking for fixed points: values  $x$  such that  $f(x) = x$ . For  $f(x) = 1 + 1/x$ , we obtain our two golden numbers. As things are set up, the number  $1.618\dots$  is an attractor with the black-hole property, while  $-0.618\dots$  is a repeller. All real inputs except zero, the numbers  $-F(n - 1)/F(n)$ , and the second golden number lead to the attractor, our black hole called phi.

## Unsolved Problems as Black Holes

Even unsolved problems sometimes fall into this black-hole scenario. Consider the Collatz Conjecture, dating back to the 1930s and still an open question (though sometimes also identified with the names of Hailstone, Ulam, and Syracuse). Start with a natural number. If odd, triple and add one. If even, take half. Keep iterating. Must you always reach 1?

If you start with 5, you get 16, then 8, then 4, then 2, then 1. Success! In fact, this problem has the paradoxical property that, although one of the hardest to settle definitively, is among the easiest to program (a few lines).

If you do reach 1 – and nobody has either proved you must, nor shown any example that doesn't – then you next get 4, then 2, then 1 again, a cycle that repeats ad infinitum. Hmm... a cycle of length three. We're interested now only in black holes, which really are cycles of length one. So, let's be creative and fix this up by modifying the problem.

Define the process instead by taking the starting number and breaking it down completely into factors that are odd and even. For instance, 84 is  $2 \times 2 \times 3 \times 7$ . Pick the largest odd factor. In the example that would be  $3 \times 7 = 21$ . (Just multiply all the odd prime factors together. The only non-odd prime is 2.) Now triple the largest odd factor and then add 1. This answer is the next iterate.

Now try some examples. You should find that you keep getting 4. Once you hit 4, you stay at 4, because the largest odd factor in 4 is 1, and  $3 \times 1 + 1 = 4$ . Anybody who proves the Collatz Conjecture will prove that my variation is a mathematical black hole as well, and conversely.

Because this is so easy to program, I will omit a program here.

## Close

Other examples of mathematical black holes arise in the study of stochastic processes. Under certain conditions, iteration of matrix powers draws one to a result that represents long-term stability independent of an input vector. There are striking resemblances to convergence results in such disciplines as differential equations, too.

Quite apart from any utility, however, the teasers and problems here are intriguing in their own right. Moreover, the real value for me is in seeing the black-hole idea as the unifying theme of seemingly disparate recreations. This is an ongoing pursuit in my own newsletter, *Recreational*

*and Educational Computing*, in which we've had additional ones. I invite readers to send me other examples or correspondence:

Dr. Michael W. Ecker, Editor  
*Recreational and Educational Computing*  
909 Violet Terrace  
Clarks Summit, Pennsylvania 18411-9206 USA

Happy Black-Hole hunting during your salute to Martin Gardner, whom we are proud to make a tiny claim to as our Senior Contributing Mathematical Editor.